
Introduction: Conventions.

Euphoria code is written in red.

Command line code and internet references are in blue.

Output is shown in green.

Software: getting started.

Euphoria is a fast interpreted language developed by Rapid Deployment Software (RDS), Ontario, Canada.

To get a copy of the software, go to http://www.rapideuphoria.com/. On the Home Page you will see what the current version is and how to download it for your operating system. Clicking on the download link will give you the basic instruction needed for an initial download.

For the purpose of this “book” I am assuming that you are using MS Windows. I am also assuming that you are using something like “M Editor” (see “Recent User Contributions” on the RDS web site, using a search for “editor”), which allows the user to key in a program and to run it from within the editor.
If you are using a different operating system then you may still find this introductory text useful but you will have to make different levels of adjustment depending upon the context.
Chapter 1: writing to the screen.

I am going to start exactly like other programming books. The first task is to write to the screen a simple piece of text, usually "Hello" or, as here, "Hello World".

In Euphoria that means writing this piece of code:

puts (1,"Hello World")

This statement, which takes the form of a procedure (or method) call, has two arguments (parameters) - the things between the brackets, separated by a comma. The procedure call tells us four things:

1. put(s) carry out the action and put out (what follows) to some output device
2. (put)s the thing which is put out is a sequence (a technical Euphoria term)
3. 1 instructs the program to display it on the default output device - usually the screen
4. “Hello World” the sequence which is to be displayed
In order to understand this a little more, I am going to write what looks like a much more complicated program.

with trace

trace(1)

sequence text

text = “Hello World”

puts (1,text)

For the moment I am assuming that you are running your programs from the “Command Prompt “ screen in Windows, or its equivalent and that you have saved this program as “hello2.ex“. Execute this program with the command:

ex hello2.ex

The first two lines of the program ensure that you enter the trace mode, akin to the debugging mode in other languages. The program’s entry point is the fourth line - simply because it is the first actual Euphoria statement. Pressing the “Enter” key steps the program into the next line. Watch carefully. The zone at the bottom of the screen produces something which at first will seem like hieroglyphics.

text={72,101,108,108,111,32,87,111,114,108,100}

It is perhaps made to look even worse by actually looking like:

text={72H,101e,108l,108l,111o,32,87W,111o,114r,108l,100d}

In fact, these extra letters give the clue. What you are seeing is how Euphoria stores what many programming languages call a “string”. It is a sequence (or “series” or “set”, in mathematical terminology) of characters, which uniquely represent the “string”.

Understand this and you understand one of the things which makes Euphoria both a simple language and a most profound one. Lots of languages have all kinds of “data types” (integers - signed, unsigned, long, short; floats - single and double; characters; strings; Boolean, etc) but Euphoria really only has one: the sequence. For completeness there is also defined an atom, which is a single (numerical) value. An object is defined as something to use when you are not sure whether you want an atom or a sequence and integer is used to represent an atom containing a (signed) integer value.

Returning to our program, then, the sequence “text” is made up of 11 atoms: 72; 101; 108: …. Each atom is represented internally by those numbers but is converted, on display, to the ASCII code equivalents, as shown by the letters: H; e: l; …

For completeness, finish tracing through the program. Another “Enter” moves you out of the “trace” screen and back to the “Command Prompt” with the program’s output - because the program has now finished.

Think about what you have been seeing and doing and about what I have said. If you have grasped all this then you are well on the way to understanding Euphoria.

Before taking a break, here is another short Euphoria program.

with trace

trace(1)

atom number1

atom number2

number1 = 8

number2 = 27

puts (1,number1+number2)

Run through the trace screen to the end, checking that it is doing what you expect. Then look at the output on returning to the “Command Prompt”. The result is

#

Were you expecting 35? Are you wrong?

Well yes, and no. Modify the program slightly, as follows:

with trace

trace(1)

atom number1

atom number2

atom result

number1 = 8

number2 = 27

result = number1+number2

puts (1,result)

Watch the trace carefully. When you get to Line 9 you will see that the variable (that is the technical term for the aliases I am using for the names in the codes) result is, indeed 35, which is translated to the ASCII code of “#” on being put to the screen! It is the nature of the puts function that displays the output so.

Chapter 2: let’s do some arithmetic?

We’ll start by going back to our last program.

with trace

trace(1)

atom number1

atom number2

atom result

number1 = 8

number2 = 27

result = number1+number2

puts (1,result)

Replace the last line with

print(1,result)

Run the program and everything is the same until you get back to the “Command Prompt”, where the result is now

35

We have used a different procedure, print, and we have got the result you probably expected in the first place.

Although technically a digression, let’s look at a slight variation of our first program:

print (1,”Hello World”)

Before running it, think what you expect to happen, including the possibility that the program may crash. Then run it and see.

The output is:

{72,101,108,108,111,32,87,111,114,108,100}

that is, the contents of the sequence. So we have one procedure puts, which treats everything as ASCII characters (in strings), and another, print, which treats everything in (internal) numeric form. Obviously we need to use the second of these if we really want to do arithmetic.

An atom in Euphoria can store pretty large numbers. These can either be integers or floating-point numbers. Signs are no problem. Fifteen decimal digits of accuracy are ensured. (Hexadecimal numbers can also be stored, but I am not planning to cover these.)

Return to our previous program and make a small change:

with trace

trace(1)

atom number1

atom number2

atom result

number1 = 8

number2 = 27.945

result = number1+number2

print (1,result)

Try this.

In following the trace you will see that, although the two numbers are technically different (one an integer, the other a decimal) Euphoria just treats them as two numbers and adds them up.

Before we leave the function puts altogether in this chapter, consider the following code:

with trace

trace(1)

atom number1

atom number2

atom result

atom a

number1 = 8

number2 = 27.945

result = number1+number2

puts (1,result)

a = getc(0)

Try running this. You will note that the contents of the variable “result” are just the same but the output is the character “#” again. So puts continues to output characters even though the value being output is a decimal – essentially it truncates first (to an integer value) and then outputs. (Note: the use of the getc() function call to hold the Command Prompt window open so that you can see the output (by pressing the “F1” key from the trace) before the window closes. If you are using an earlier version of MS Windows this trick may not be needed.)

I suggest you now do lots of arithmetical examples of your own. As you get more confident you can drop the trace commands, and instead use the getc() convention to study your output. I suggest you including signed numbers (called unaries in the Euphoria manual), consider both integers and floating-point numbers and, if you want, some hexadecimal arithmetic.

Before we leave the subject of arithmetic altogether, I want to touch upon how the arithmetic operators (+, -, * and /) apply when the object is a sequence and not just an atom.

Suppose we have the following sequence:

{3,12.6,44,5e2}

What happens when we, say, try to double it all?

Consider the program:

sequence x,y

x = {3,12.6,44,5e2}

y = 2*x

atom a

print(1,y)

a = getc(0)

You will notice that each of the elements of “x” is doubled.

Now try:

sequence x,y

x = {3,12.6,44,5e2}

y = {4.6,3.1e1,-56,19,32.78}

atom a

print(1,x+y)

a = getc(0)

You get an error, filed in “ex.err”, which tells you that the two sequences are of unequal length.

Before we leave this subject, let's take a double look forward. Consider the following program:

atom a,i

i = 14

for j = 1 to 10 do

    ? i/j
    puts (1,10)

end for

a = getc(0)

There are some new little “tricks” in this program but it is not why I have put this program here. The main purpose is to show why we need to introduce a further output procedure. But we need to take a look at these new features first.

We take an advanced look at a looping structure, I show you a shorthand way of calling the print procedure and a rather coded way of forcing a “new line”.

Looking at the output, you will notice that the results of the calculations have different numbers of decimal places, in accordance with the nature of the division. Consequently the output layout is less than ideal. If you want to get consistency of output for numeric values then you need to use a more versatile function than print. 

Try this alternative program:

atom a,i

i = 14

for j = 1 to 10 do

    printf (1,”%8.3f”,i/j)

    puts (1,10)

end for

a = getc(0)

The function printf is a much more sophisticated output procedure. This is because it takes an extra parameter (the second), which indicates the “format” the output should take. In our case this is a floating-point number occupying eight characters, three of which are after the decimal point.

There are eight format types, covering all data “types”, including strings. The format value can also be negative. It can be used to “pad” a numeral with leading zeros and include a “+” sign against a positive numeral. Format codes can be combined, if sequences of more than one element are to be printed together, and can carry escape sequences. The final version of this program omits the puts statement and includes the “line feed” escape sequence (\n) in the format. It also shows how the negative format yields left-justification.

atom a,i

i = 14

for j = 1 to 10 do

    printf (1,”%-8.3f\n”,i/j)

end for

a = getc(0)

Chapter 3: more about sequences.

Up to now the sequences you have dealt with have been series of atoms. Strictly speaking, this is not a definition. A sequence is actually a series of objects.
First of all, though, perhaps we should register the existence of the “null” string, shown as {}.

Take now a sequence of strings, as exemplified in the example:

sequence aSequence

aSequence = {“John”,”Smith”}

puts (1,aSequence)

atom a

a = getc(0)

Try this.

You get an error message because puts only takes a single argument, which our sequence isn’t.

Try instead:

sequence aSequence

aSequence = {“John”,”Smith”}

puts (1,aSequence[1])

puts (1,aSequence[2])

atom a

a = getc(0)

Now it is OK, but this again shows the limitation of the puts statement.

What, you might wonder would happen if we tried print instead.

sequence aSequence

aSequence = {“John”,”Smith”}

print (1,aSequence)

atom a

a = getc(0)

You have probably guessed that print recognizes all elements of the sequence but prints them out in elemental form.

What about printf? Well it is both straightforward and not.

Look first at this program:

sequence aSequence

aSequence = {“John”,”Smith”}

printf (1,”%s”,aSequence)

atom a

a = getc(0)

You might think this would produce the output:

JohnSmith

but it doesn’t. What does work is something like this:

sequence aSequence

aSequence = {“John”,”Smith”}

printf (1,”%s %s”,aSequence)

atom a

a = getc(0)

which recognizes the two-dimensional structure of the sequence.

Contrast this with:

sequence aSequence

aSequence = {“John Smith”}

printf (1,”%s”,{aSequence})

atom a

a = getc(0)

where the name resides within a single component of the sequence structure.

It is important that you recognize the difference between this and:

sequence aSequence

aSequence = {“John Smith”}

printf (1,”%s %s”,aSequence)

atom a

a = getc(0)

What these simple (though not easy) examples illustrate is that sequences are like arrays (or dimensions) in other programming languages. In the sequence:

{“John Smith”}
the first element is the character “J”, the second “o”, the third “h”, etc. In contrast, in the sequence:

{“John”,”Smith”}
you need two dimensions to describe the elements. So if this sequence is assigned to the variable called aSequence then aSequence[1,1] is “J” and aSequence[1,4] is “n”; just as aSequence[2,1] is “S” and aSequence[2,3] is “i”. In the same way, aSequence[2] is “Smith”.

This arrangement within Euphoria, although needing time to absorb properly, is extremely powerful and flexible. It is very useful for constructing what are known as “data structures”.

Suppose you want to set up a database of people who are members of a club. You want to record (at least) the following items of data:

· Name

· Address

· Category of membership

· Membership number

· Date of renewal

Each data item can be further “sub-divided”. For example “Name” can be made up of:

· Title

· First Name

· Initials

· Surname

Addresses tend to be rather difficult things to structure, especially in country regions. I would probably just break it down into separate “lines”. Dates are also a minefield but can easily be managed, for this application, with sub-categories of:

· Day

· Month

· Year

A relevant arrayed structure could then be: 

members = {

                  {{“Mr”,"John",”J.S.”,"Smith"},

                  {“Appletree Cottage”,”12 Back Lane”,”South Coker”,”Yeovil”,”BA22 7YZ”},

                   “Full”,

                   1456,

                   {23,04,2006}},

                  {{“Ms”,"Jane",”J.T.”,"Smythe"},

                  {”12 London Road”,”Blackpool”,”FY1 1AA”},

                   “Retired”,

                   1105,

                   {01,01,2007}}

                   etc

                }

If we then declared a few constants, values which cannot be changed, we could make cross-referencing fairly explicit. For example:

constant NAME = 1

constant TITLE = 1, FIRST_NAME = 2, INITIALS = 3, SURNAME = 4

constant ADDRESS = 2

constant MEMBERSHIP = 3

constant MEMBERSHIP_NUMBER = 4

constant RENEWAL = 5

constant DD = 1, MM = 2, YYYY = 3

would mean that 

members[2][NAME][SURNAME] is “Smythe”

and

members[1][RENEWAL][YYYY] is 2006.

Here is the basis of a little report:

puts (1,"Membership Renewals")

puts (1,"\n\n")

for member = 1 to length(members) do


printf (1,"%-3s ",{members[member][NAME][TITLE]})


printf (1,"%-5s ",{members[member][NAME][INITIALS]})


printf (1,"%-15s ",{members[member][NAME][SURNAME]})

            printf (1,"(%05d) ",members[member][MEMBERSHIP_NUMBER])


printf (1,"%-10s : Renewal = ",{members[member][MEMBERSHIP]})


printf (1,"%02d/",members[member][RENEWAL][DD])


printf (1,"%02d/",members[member][RENEWAL][MM])


printf (1,"%4d",members[member][RENEWAL][YYYY])


puts (1,"\n")


end for
My advice is to take as long as you need to work through lots of examples like these, from simple to more complex. You will make mistakes, I imagine, unless you are very smart, but persevere and you will have made another huge step in understanding Euphoria.

Chapter 4: yet more about sequences.

In the last programming illustration I used length(), one of the standard built-in functions that can be used on sequences. This function takes as its argument a sequence. So let’s think about what this means?

What do you think is the answer to length(“John Smith”)? There are logical arguments (in the general sense) for 1, for 2 and for 10. Make sure you understand why, in Euphoria, the answer is 10.

I am sure that you will now easily solve length({“John”,”Smith”}) and also see why, in the “database” example at the end of the previous chapter, despite all those data fields, the same answer applies there too.

Sometimes you want to create a sequence with values already set. If those values are not too complicated then the repeat function may be useful. This takes two arguments:

· a value

· the number of times the value should be added to the sequence

The “value” can be an atom or a sequence; the “number” can be any integer value, even zero, In that case, however, nothing will be added.

There are some instances where you might want to create a sequence with a mixture of initial values. Let us take a simple case: we want 25 instances of “apple” and 50 of “orange” in our sequence. Here is one way to do it.

sequence apple

sequence orange

sequence fruit

apple = repeat("apple",25)

orange = repeat("orange",50)

fruit = apple & orange

for index = 1 to 75 do


printf (1,"%s ",{fruit[index]})

end for

atom a

a = getc(0)

The one new operator used here is &, the concatenate operator, which puts the second sequence on the end of the first. 

It is possible to illustrate the two other sequence-based functions by solving the same problem twice more. Take a look at this little program:

sequence fruit

fruit = repeat("apple",25)

for index = 1 to 50 do


fruit = append(fruit,"orange")

end for

for index = 1 to 75 do


printf (1,"%s ",{fruit[index]})

end for

atom a

a = getc(0)

This uses the function append to add the string “orange” to the end of the sequence fifty times.

Now look at this next little program:

sequence fruit

fruit = repeat("orange",50)

for index = 1 to 25 do


fruit = prepend(fruit,"apple")

end for

for index = 1 to 75 do


printf (1,"%s ",{fruit[index]})

end for

atom a

a = getc(0)

This shows how the function prepend works.

Chapter 5: more about statements.

In Euphoria there are eight types of statement. These are:

· assignment statement 

· if statement 

· while statement 

· for statement 

· exit statement
· return statement 

· procedure call
· function call 
In this chapter we will look further at the first five of these types and illustrate how they are used. The next chapter will deal with the others.

At the most basic level an assignment sets the value of a variable. We have seen a few of these already in the earlier programs. I have collected them all together here (Note: this is a list and not a program!):

text = “Hello World”

number1 = 8

number2 = 27

number2 = 27.945

i = 14

x = {3,12.6,44,5e2}

aSequence = {“John”,”Smith”}

aSequence = {“John Smith”}

members = {

                  {{“Mr”,"John",”J.S.”,"Smith"},

                  {“Appletree Cottage”,”12 Back Lane”,”South Coker”,”Yeovil”,”BA22 7YZ”},

                   “Full”,

                   1456,

                   {23,04,2006}},

                  {{“Ms”,"Jane",”J.T.”,"Smythe"},

                  {”12 London Road”,”Blackpool”,”FY1 1AA”},

                   “Retired”,

                   1105,

                   {01,01,2007}}

                   etc

                }

result = number1+number2

fruit = apple & orange

y = 2*x

a = getc(0)

apple = repeat("apple",25)

orange = repeat("orange",50)

fruit = append(fruit,"orange")

fruit = prepend(fruit,"apple")
The first group are simple assignments. Here a variable, already defined by type, receives a value (an atom or sequence according to type).

In the second group, a variable’s value is determined by the value of another variable or a combination of variables.

The third group receive values by calling functions (covered in more detail in the next chapter).

The essential aspect of all assignments, however, is that a variable receives the value. That value is determined by the properties of the right-hand side of what looks rather like a mathematical equation. Assignments are the fundamental building block of all Euphoria programs. If several assignment statements follow one another then the order in which they appear determines their order of execution. Put simply, consider these two pieces of similar code:

x = 7

y = 21

x = y +3

y = x*2

x = 7

y = x*2

x = y +3

y = 21

In the first series of assignments the variable x ends up as 24 and y as 576. In the second series of assignments the variable x ends up as 52 and y as 21. Despite all the assignments being the same, the order of execution is the prime factor is determining the result.

Traditional programming is sequential in nature (see later for a consideration of Windows-based programming which essentially isn’t). However a language needs more complex structures than assignments to address real problems. All the remaining statement types considered in this chapter are either more complex structures or elements within structures which, in one way or another, have the potential either to change the order in which statements are executed within the structures or whether the statements within the structure are executed and/or re-executed.

Now let’s take a look at the programming structures Euphoria offers.

The first is called the “if” statement. Its simplest form is:

if condition then
    statement(s)

end if
where “condition” is an expression that is either “false” (0) or “true” (non-zero). The “statement(s)” are one or more assignments or other programming structures which are to be executed if the condition is true.

A more complicated version is:

if condition then
    statement(s)

else

    statement(s) 

end if
Here the second set of statements is executed if the condition is false.

There are yet more complicated versions of this structure, but I will not cover them here, because we need to consider the nature of conditions with some care at this point.

The simplest (but least useful) form of condition is given by a single value. For example the condition 3 is always true, and the condition 0 is always false.

The next layer of sophistication involves comparison. For this we need what are known as the “comparison operators” (=, >, <, >=, <=, !). So, for an integer, x>5 is true for all values of x from 6 upwards, whereas x>=5 is true for all values of x from 5 upwards. The equality (=) is no doubt obvious, but != may be less so. In fact it is shorthand for “not equal to”!

It is possible to have compound comparisons by use of the keywords and and or. Examples of compound comparisons are:

char = 'b' or char = 'B'

x != 0 and length(nameString) > 5

Here is a simple example of the use of the if statement:

sequence name

puts (1,"Do you want me to write out your name? ")

object response

response = gets(0)

puts(1,'\n')

if (response[1] = 'y') or (response[1] = 'Y') then


puts(1,"Well what is your name then? ")



name = gets(0)


name = name[1..length(name)-1]


puts(1,'\n')


puts(1,"Hello, " & name)

else


puts(1,"Well, it was your choice not mine!")

end if

atom a

a = getc(0)

Strictly speaking this is not the best way of doing this, but it shows a compound comparison in action. Although not part of your understanding of the if statement, there are some new ideas here: using the gets function; stripping the “return” off the end of a string so “got”; forcing a “return” to the screen after you have called the function. Although the second is optional here, it is good practice in case you want to store the string elsewhere for later use. Try commenting out Line 10 (put the code “--“ in front of the line) and running the program. Interesting? Can you see why you get this effect?

The while statement in Euphoria enables a block of code to be executed repeatedly, until a set condition changes.

Here is a simple example:

atom i

i = 0

while i < 10 do


printf (1,"The counter is %d\n",i)


i += 1

end while
atom a

a = getc(0)

Note that the while loop could go on indefinitely if the condition remains “true”. So as the programmer, you should either change the circumstances inside the loop, as in this example, or use as a condition something which changes indirectly as a result of what you are doing inside the loop.

(Note a new, but very useful, little coding device: i += 1 is shorthand for i = i + 1.)

A good example of the latter occurs when reading a file. We are going to read the contents of a file using the function getc. This function reads characters (bytes) one at a time. When the file reaches its end the atom -1 is returned. This program illustrates the approach:
atom FileHandle

atom c

atom lines

atom characters

sequence FileName

lines = 0

characters = 0

puts (1,"What file do you want to scan? ")

FileName = gets(0)

FileName = FileName [1..$-1]
FileHandle = open(FileName,"r")

while 1 do


c = getc(FileHandle)


if c = -1 then



exit


end if


if c = '\n' then



lines += 1


else



characters += 1



end if

end while

printf (1, "\nFile \"%s\" contains ", {FileName})

printf (1,"%d lines and ", lines)

printf (1,"%d characters\n", characters)

atom a

a = getc(0)
The main thing to notice is the use of while 1 do. The value 1 is Euphoria-speak for “true”. So the loop theoretically goes on for ever. What we do is test for “end of file” (a return value from getc of “-1”) and jump out of the loop, using the command exit. (Note this is another of the eight Euphoria statements introduced at the start of the chapter.)

There are a few other (new) things to notice about this little program.

· The trimming of the file name (otherwise there is a file error, as the gets function returns the \n at the end of the line). Note also the use of the code $ as shorthand for length(sequence). This is a feature only of the newest edition of Euphoria, so don’t use it if you already have an older version.
· The exit statement effects immediate termination of the loop. The program jumps to the next statement after the end of the loop. It can also be used inside a for loop – see below.

· The use of the escape character \” to enable you to include double quotes inside a quoted sequence.

· What the program counts as a valid character to include in the overall count. We have trapped “line feed” but not carriage return. Nor have we distinguished between blank characters and ones which form parts of “words” or “formulae”.
This last point raises two interesting aspects about programming. One is general and one specific to Euphoria.

1. My programs have been chosen as illustrative and have been made as simple as possible for you to read and understand. When writing real programs one needs to be very sure exactly what you are trying to achieve. In this case are we only interested in the length of the file (an easier program than the one above) or should we be trying to dissect the file more and count a wider variety of different characters, possible including counting the non-printable ones separately?
2. Many programming languages have a “case” or “switch” construct defined, in which it is possible to test a value (say a character) and list a whole array of possibilities and resulting actions within the construct. In Euphoria there is no such construct. You have to use nested if statements, so to get the same effect you may have considerable levels of nesting.
I have already used for loops in earlier illustrations, so I will concentrate here on the details of the statement rather than its illustration.
The full form of the statement is:

for loop-variable = initial-value to final-value by step do


… statement(s)

end for

The loop-variable is defined from within the loop, so you do not need to declare it first.

The loop executes by setting the loop-variable to the initial-value. It compares the loop-variable with the final-value and provided it is not higher then it carries out the loop, then adds the step value to the loop variable. The process repeats until the loop-variable exceeds the final-value.
The initial-value, final-value and step are all atoms, but can be either integers of decimals. They can be positive or negative, so you have to be responsible for ensuring that the loop can end. Remember, however, that the exit statement can be used inside a for loop to effect immediate closure.
If you leave out the step element of the statement, then a value of “+1” is assumed.

If, for some reason, you want to capture the value of the loop-variable for use outside the loop, then you have to declare a (atom) variable and assign the loop-variable value to that variable at the appropriate point in the loop. If you need to know more about the nature of the loop-variable, then read the section “scope of the loop variable in 2.4.2 Scope.” in the Euphoria manual. As yet, you haven’t needed to understand how long, or under what conditions, a variable exists (its “scope”) but we shall need to study this matter seriously in the next chapter.
Chapter 6: parts and wholes

Some computer languages require that, for an application to run, there must be a “main” program. Euphoria is not, strictly, such a language but it adheres quite closely to the principles behind such an arrangement. The way in which this shows is through the naming conventions of programs and other parts (which we are about to explore!).

Actual applications (working programs) have extension either of the form “.ex” (for Console-based) or “.exw” (for Windows-based).

If you see Euphoria code with the extension “.e” then the file is not an application but is what is known as an “include” file. It is either Console-based or of general applicability. Files with the extension “.ew” are Windows-specific include files (that is with code which will only work within a Windows-based environment).

The purpose of this chapter is to look beyond these conventions and explore how programs can be constructed, both using already-written pieces of code and by using devices (procedures and functions) which simplify repetitive elements of a program.

So far we have used a few of the standard functions and procedures provided by Euphoria. Below I list those so far encountered.

puts

trace

print

printf

getc

length

repeat

append

prepend
Why have I put them into two groups? Well, the first group are all procedures and the second group all functions. What is the difference?

The main difference between a procedure and a function is that the latter returns a value and so we tend to see functions appearing on the right-hand side of assignment statements. Procedures never can so appear. Why did I say “tend to” rather than “always”? The reason is because there are times (we have used two already) when the assignment is implicit.

Take a simple example:

atom x

sequence string

string = “Hello, World!”

x = length(string)

printf (1,“The length of the string is %d characters”,x)

atom a

a = getc(0)

Now look at the following:

sequence string

string = “Hello, World!”

printf (1,“The length of the string is %d characters”, length(string))

atom a

a = getc(0)

These are essentially identical programs, but in the second, because we never wish to use the variable x again, we can use an implicit reference to an atom in the printf statement.

Can you spot the other instance of implicit use of a function? The statement was:

for member = 1 to length(members) do

in the report in Chapter 3.

Functions and procedures (I will call them “routines”, collectively) can be located in one of four places. I will call these places “libraries”.

The first group are defined within the Euphoria language and are built-in to the language. You just call them from within your program as and when you need them. The library is internal.

There is a second group of routines, which are also defined within the language but are stored in supplementary libraries. Before your program can access these you must prefix your program with an include statement. Immediately after the statement is the name of the library you wish to call.

For example, there are functions lower and upper which can change the case of a text. They are located in the wildcard library. Here is a simple use of these functions

include wildcard.e

sequence string

string = "Today is Friday"

puts (1,string & '\n')

puts (1,lower(string) & '\n')

puts (1,upper(string))

atom a

a = getc(0)

A third source is material which someone else has written and made public. As a general rule I would always avoid writing something afresh if someone else has done it before you. Assuming, of course, that you know about it! I offer two general pieces of advice here:

· look at your Euphoria Programming Language Reference Manual for functions or procedures which might cover your needs

· before embarking on a serious piece of programming look at the RDS web site under “User Contributions”, using the search engine, to see if there already is something that you could utilise or modify for your needs

To illustrate this point, look again at the last illustration. Take a special look at the last line. Remember we put this in purely to stop the Command Prompt window closing before we have perused the output. For someone new to a program like this, however, the programming is too terse: it doesn’t tell the user what to do. So look at this alternative:

atom a

puts (1,“Press RETURN to close the window”)

a = getc(0)

I think you will agree that it is clearer now.

I could, instead, have used a built-in function to do this job for me. The function prompt_string, in the include library get, returns a sequence. Its argument is a string which displays on the screen. Here is the alternative program.

include wildcard.e
include get.e

sequence string

string = "Today is Friday"

puts (1,string & '\n')

puts (1,lower(string) & '\n')

puts (1,upper(string) & '\n')

sequence result
result = prompt_string(“Press RETURN to close the window”)
Admittedly in this case the function is more powerful than the need – we don’t use the return value, but it is clearly a very useful thing to have when setting up a user dialogue. Moreover, there is a companion function, prompt_number, which does the same thing but returns an atom. It has the additional facility of providing boundary values for the number. Here is a simple example, which shows off both functions:

include get.e

sequence name

name = prompt_string("What is your name? ")

atom age

age = prompt_number("How old are you? ",{3,105})

puts (1, "Hello, " & name)

printf (1,"\nYou are %d",age)

sequence response

response = prompt_string("\nPress RETURN to close the window")

Try this program a few times with different values. Note that it doesn’t matter how much detail you put into your name – it just stores and repeats it. Try entering an out-of-range age.

In the Reference Manual, under prompt_number, it states:

“If this routine is too simple for your needs, feel free to copy it and make your own more specialized version.”
Don’t dismiss this comment. As you get more expert then the idea of modifying code for you own needs becomes increasingly possible, as well as attractive.

Of course, as you get more experienced you can create your own libraries. I shall show you how easy this is shortly, but let’s first cover the fourth instance: routines can be stored within the program itself. You would do this if you had an operation you wanted to perform several times within the program as a whole and could thus save repeating similar blocks of coding.

I’ll give you three examples of writing functions and one of a procedure. First we’ll put then inside our calling programs, then when all are written then we’ll put them into libraries and call them ourselves.

My first example is a function to square a number. The code is pretty easy, but take a close look at the detail.

function square(atom number)


return number*number

end function
Line 1 defines the function name and, within the parentheses, the one or more arguments it takes. The last line, here Line 3, defines the end of the coding for the function. In between are the statements which compute the value(s) the function is dealing with. This section must include at least one return statement, which gives the value the function offers to the calling program. So let us now embed the code inside a little program:

include get.e

function square(atom number)


return number*number

end function

for i = 1 to 6 do


printf (1,"%d\n",square(i))

end for

for i = 1 to 2 by 0.1 do


printf (1,"%4.2f\n",square(i))

end for

sequence response

response = prompt_string("\nPress RETURN to close the window")
Note that the function must exist before it is called. Note, too, that the function works for both integral and decimal values. You may think this is an obvious point, but in many languages you would need a separate function for each type of number that the language defines.
Before leaving this example, try a small change. Make the function’s argument into an object instead. This is a Euphoria type that allows either an atom or a sequence to be used. Try this slightly modified program:
include get.e

function square(object number)


return number*number

end function

sequence run, squarerun

run = {2,3,4}

squarerun = square(run)

for i = 1 to length(squarerun) do


printf (1,"%d\n",squarerun[i])

end for

for i = 1 to 2 by 0.1 do


printf (1,"%4.2f\n",square(i))

end for

sequence response

response = prompt_string("\nPress RETURN to close the window")
Study the result. You’ll see that, whilst it still works perfectly for the decimal values, when a sequence is input as the argument, all the elements of the sequence are squared. If, like me, you are a statistician, then you just love Euphoria at times like this!
Another simple function is to find the maximum of two values. The following does this:

function max(atom x, atom y)

if x > y then


return x

else


return y

end if

end function

Try it, with any values you want, by writing a little program to call the function.

In the light of our previous example, we could have a different sort of “maximum” function. Look at this code:

include get.e

function max(sequence x)

atom result

result = 0

for i = 1 to length(x) do


if x[i] > result then



result = x[i]


end if

end for

return result

end function

sequence row

row = {1,2,3,5,6,9,4}

printf (1,"The maximum of the row of numbers is %d",max(row))

sequence response

response = prompt_string("\nPress RETURN to close the window")
(Note: this even works if the sequence only has one member!)

It pays to think about precisely how you might want to organise your data before jumping in and writing routines. In this case it is quite difficult to say which might prove the more useful, but the second version is certainly more general.

My last example of a function involves a new technique, so keep up the concentration level!

The mathematical function “factorial” is defined as follows:

For positive integer x, if x = 1 then factorial(x) = 1,

otherwise factorial(x) = factorial(x – 1) * x

This sort of definition is called “recursive”. Some programming languages allow routines to be recursive, but others do not. Look at this function definition in Euphoria:

function factorial (atom number)

if number <= 0 then


return -999

else


if number = 1 then



return 1


else



return number*factorial(number - 1)


end if

end if

end function

include get.e

for i = 1 to 10 do


printf (1,"Factorial %d ",i)


printf (1,"is %d\n",factorial(i))

end for

for i = 1 to -2 by -1 do


printf (1,"Factorial %d ",i)


printf (1,"is %d\n",factorial(i))

end for

sequence response

response = prompt_string("\nPress RETURN to close the window")

Try this and you see that, indeed, Euphoria is one of the languages that supports recursion.
Note that, in this case, I have generated an “error-number” code (-999) to cover instances where the value passed to the function makes no sense. An important part of programming is the identification and trapping of error conditions.

Are you getting fed up of keep having to add the code to stop the Command Prompt window closing? So am I. Well here is a simple way around this:
include get.e

procedure pause()

sequence response

response = prompt_string("\nPress RETURN to close the window")

end procedure

puts (1,"Hello, there!")

pause()
If we were to put this procedure inside a library of our own, say utilities.e, then in any program all we would need to do is put

include utilities.e

at the start of the program and then we can write

pause()
at any time!

Next we will put our routines together into libraries of our own. Take the three mathematical type functions and put them sequentially into a file.
function square(object number)


return number*number

end function

function max(sequence x)

atom result

result = 0

for i = 1 to length(x) do


if x[i] > result then



result = x[i]


end if

end for

return result

end function

function factorial (atom number)

if number <= 0 then


return -999

else


if number = 1 then



return 1


else



return number*factorial(number - 1)


end if

end if

end function
Save this file as, say, maths.e. (I’m English!)
I am also assuming that you have put the “pause” procedure in a file called utilities.e.
Now test it out. For example:

include maths.e

include utilities.e

sequence run, squarerun

run = {2,3,4}

squarerun = square(run)

for i = 1 to length(squarerun) do


printf (1,"%d\n",squarerun[i])

end for

for i = 1 to 2 by 0.1 do


printf (1,"%4.2f\n",square(i))

end for

sequence row

row = {1,2,3,5,6,9,4}

printf (1,"The maximum of the row of numbers is %d\n",max(row))

for i = 1 to 10 do


printf (1,"Factorial %d ",i)


printf (1,"is %d\n",factorial(i))

end for

pause()
When it runs you get an error message which tells you that the first function you called, here square, has not been defined. The issue here is known as “scope”. I’ll give you the solution first, then the explanation.

In front of all your routine declarations you key the term global. Re-save the libraries and try this program again. It now works fine.

Scope is quite a big subject in computing. I do not claim to cover it fully here, but I hope to say enough for you to get by. Essentially it covers the life-span of all programming constructs: variables, routines and user-defined types (not covered in this introduction). By default all constructs have scope only in the program, or routine, in which they are declared. On balance, this is the best possible world, otherwise you could have, by accident, all sorts of horrible, even disastrous, naming conflicts. The down-side, however, is that, if you want someone to use your library routines, including yourself, then all the routines, variables and types you include in your library must be set to global.
The following program might help in understanding.

include utilities.e

function callname(sequence name)


sequence fred


fred = "Hi, Fred!"


puts (1,fred & '\n')


return name

end function

procedure farewell()


sequence fred


fred = "Bye, Fred!"


puts (1,fred & '\n')

end procedure

procedure printit(sequence name)


printf (1,"The sequence in the main program is %s\n",{name})

end procedure

sequence fred, reply

fred = "Hello, Fred"

printit(fred)

reply = callname(fred)

printit(fred)

farewell()

printit(fred)

pause()
Study the code and see if it helps you to understand about scope.
Chapter 7: error handling

Even if you have tested out all my programs and not made any errors in transcribing or copying them, you will already have met the file “ex.err”. In the Command Prompt Window it pops up as errors occur; when using MEditor it also opens automatically after returning from the (failed) execution. This file documents the error(s) trapped at the point at which Euphoria stops processing the program. If you are calling a library then the error point could be within the library, so, if it not one you wrote, the error message may make no sense to you. The error log, however, does at least indicate the point in the calling program which led to the error. If you call a library, it calls another one, etc., then the error trace may be quite lengthy, but you can always trace it back to the “main” program which initiated the chain of events.
We have already covered, at the start, the use of trace. I would always advocate using this facility when you are trying anything new. Life also is much easier if you break down any complicated programming task into smaller parts, so that each can be checked separately. (Programs written this way are always easier to modify in the future, too.)

Whilst not going into detail, as well as “ex.err”, there is also a file called “ctrace.out”, which you can open to get a full log of programming steps. Look at the Euphoria Reference Manual for full details. Essentially you get this option when you escalate the trace level.

When writing your own routines, it always pays to think about errors that might be made when calling them. The most common problem is calling a routine with the wrong kind of argument. Look at this example:
include maths.e

include utilities.e

sequence wrongnumber

wrongnumber = {12}

print (1,factorial(wrongnumber))

pause()

When you try to run it Euphoria comes up with a “type_check failure, number is {12}”.

When deciding to write a new routine one consideration is error trapping. A return value in a function can be used to report back whether an error has occurred or not.

First here is a simple procedure:

procedure openfile(sequence filename)


atom FileHandle


FileHandle = open(filename,"r")

end procedure
This works, in the sense that the open function is correctly called but it is limited in at least two ways. Now here is the same code, packaged as a function, with a return code for testing.

function openfile(sequence filename)


return open(filename,"r")

end function
If the function returns -1 then the file has not been found or doesn’t exist.

If possible, make the return code zero for an error, as the condition “false” is represented by zero.
Chapter 8: simple Windows programs

So far, apart from one slight reference, all our examples have been “DOS”-based, that is, have run within a Command Prompt Window. There is no reason, however, why you should not now try to understand something about Windows and how the Euphoria language can be used to respond to that different environment.

First, however, you need to consider how these two environments are different.

Traditionally programming has followed the sequential model. The first statement is executed, then the second, … to the end of the program. Unfortunately there were times when it was necessary to jump forward to miss out some statements and even to jump back to an earlier point. Critics called this kind of programming “spaghetti”. Structured programming techniques were developed to overcome this weakness and we have seen above how Euphoria’s structures allow the sequential model (structure-to-structure) to continue to be implemented.

A Windowing environment (not just MS Windows) is totally different, as you will probably have appreciated from your own use of one. Here things (windows, and within windows all sorts of “widgets” like buttons, labels, menus, etc) are created and sit on your desktop. Essentially they are all potentially active but waiting for some initiative to activate them. Often that initiative comes from the user (clicking a mouse button, using the keyboard, etc), which may well activate a chain of subsequent operations (selecting a menu item may well open another window or ask you to enter some text, etc) but it can come from other processes, because MS Windows is a multi-tasking operating system.

You can find a range of vocabulary in different books to describe these features. I will choose to call all the things “objects”. (Slightly unfortunately, Euphoria has a definition of object which is not at all the same. I have tried to avoid using the term hitherto unless I had to!)

Objects sit, in a kind of limbo, awaiting “messages”. Each extant object is open for the receipt of messages and the operating system directs to the relevant object(s) any which are issued. Objects, in general, are also capable of issuing messages. Remember that users are, in this sense, objects, too.

Before we move on to look at how Euphoria can respond to this different environment, we need a brief history lesson in how MS-DOS became MS Windows. 

In order for Windows programs to be written, Microsoft constructed a set of tools, known as the API (Application Program Interface). At the time of its construction the most common programming language was C and so the Windows API was written in this language. As time went on providers of other languages devised ways for their users to call the routines within the API without having to learn, or use, C. Euphoria was no exception and the “win32lib” library was born. (There are other options for writing windows-based programs in Euphoria, but this is the most commonly-used resource at this time.)
Writing programs in the API has never been easy. For example, to create a window object with no functionality takes 60-70 lines of code, in C. It doesn’t take quite so many steps in Euphoria!

Here is that simplest of programs – a blank window!

include win32lib.ew

constant

BasicWindow = create (Window,
"Basic Window",0,Default,Default,
Default,Default,0)

WinMain (BasicWindow,Normal)

Dissecting this program is not too difficult. The first line calls the library which enables Euphoria code to signal API messages. The second line (spread across four for visibility) defines a constant, with the name “BasicWindow”, which is created with set parameters (remember constants don’t change). The first parameter specifies the kind of Object it is, the second its Title (sometimes called a “Caption”), the third identifies its structural position in the hierarchy of objects (here the zero implies it is at the top level). The next four parameters indicate position (Left, Top, Width, Height); here we adopt the default values. The final parameter defines the type of object we want; again zero is the default – for a plain window. The third line of the program calls the main processing loop for win32lib. This takes two parameters: the first is a sequence, the second an atom. The first in this case just gives the name of the main window. The second identifies the “style” in which the window will be opened. Here the value “Normal” signifies that the window should be opened with the dimensions specified in the create statement.

I now give a slightly more complicated example to show you how you can build a working windows program. I have put in some spacing to improve readability.
include win32lib.ew

constant

DialogWindow = create (Window,
"A Window with a button", 0,     Default,Default,200,100,{WS_DLGFRAME}),
 

BasicButton = create (Button,
"Press me",DialogWindow,50,10,75,20,0)

global procedure onClick_BasicButton()

           closeWindow( DialogWindow )

end procedure

onClick[BasicButton] = routine_id( "onClick_BasicButton" )

WinMain ({DialogWindow,BasicButton},Normal)

Here we create two objects: a window and a button placed on it. Note that the third parameter in the button-create statement indicates that the button belongs to the window (the window is its “parent”). This time I set some sizes to both the window and the button, but I still let the system position the window where it chooses. I also introduce a “flag”, in the form of a one-element sequence, so that the window cannot be closed, re-sized, minimised or maximised. This is known as a “dialog”-style window.

In order to make the button do something we need to define a procedure for each action we want it to perform – in this case what to do if it is “click”ed by the left-hand mouse button. In this case we ask it to close the (parent) window, as we have disabled that option in the parent window itself.

We also need to create a link between an event (here clicking a button) and the procedure to handle that event.

Finally we call the main processing loop. It looks nearly the same as last time but here the first parameter not only passes the name of the main window but also indicates what object should get the “focus” when the application is opened (not really too relevant for such a simple case, but explaining another feature on the way).

I will end with a rather more complex example. I won’t give you any commentary, so that you can think about it yourself.

include win32lib.ew

include msgbox.e

constant

MainWindow = create(Window, "A final example",0,
  Default, Default, 800, 500, 0),


FileMenu = create(Menu,"&File",MainWindow,0,0,0,0,0),


OpenMenu = create(MenuItem, "&Open", FileMenu,
         0, 0,0, 0, 0),


PrintMenu = create(MenuItem, "&Print", FileMenu,
         0,0,0,0,0),


SpacerMenu = create(MenuSpacer, "", FileMenu,
         0,0,0,0,0),


ExitMenu = create(MenuItem,"E&xit",FileMenu,0,0,0,0,0),


HelpMenu = create(Menu, "&Help", MainWindow, 

        0,0,0,0,0),


AidMenu = create(MenuItem, "&Help", HelpMenu,
        0,0,0,0,0),


AboutMenu = create(MenuItem, "&About", HelpMenu,
        0,0,0,0,0),


EditZone = create(RichEdit, "", MainWindow,
        Default,Default,775,440,0)

sequence


FileName,


Buffer

global procedure onOpenMenu_Select()


atom Character


atom FileHandle


FileName = getOpenFileName(MainWindow,"",

           {"Text File", "*.TXT","All Files", "*.*"} )


FileHandle = open (FileName,"r")


setFont(EditZone,"Courier New",12,{"ALL",Normal})



if FileHandle != -1 then



Buffer = {}



Character = 0



while Character != -1 do




Character = getc( FileHandle)




Buffer &= Character



end while



close (FileHandle)


end if


Buffer = Buffer[1..length(Buffer)-1]


setText(EditZone,Buffer)


end procedure


onClick[OpenMenu] = routine_id( "onOpenMenu_Select" )

global procedure onPrintMenu_Select()


sequence result,



    fsize


atom response,


         x,


         y


result = getPrinter()


setFont(Printer,"Courier New",10,Normal)


fsize = getFontSize(Printer)


x = 0


y = 0


if length( result ) then



response = startDoc(FileName) 



wPuts (Printer,{"File: %s",{FileName}})



setFont(Printer,"Courier New",12,Normal)



y += 2*fsize[2]



setPenPos (Printer,x,y)



for i = 1 to length(Buffer) do




if Buffer[i] > 13 then





x += fsize[1]





setPenPos (Printer,x,y)





wPuts (Printer,{Buffer[i]})




elsif Buffer[i] = 10 then





y += fsize[2]





x = 0





setPenPos (Printer,x,y)




end if




end for



response = endDoc()



releasePrinter()


end if

end procedure

  onClick[PrintMenu] = routine_id( "onPrintMenu_Select" )

global procedure onSpacerMenu_Select()


-- dummy to avoid error message

end procedure

  onClick[SpacerMenu] = routine_id( "onSpacerMenu_Select" )

global procedure onExitMenu_Select()


closeWindow( MainWindow )

end procedure

  onClick[ExitMenu] = routine_id( "onExitMenu_Select" )

global procedure onAidMenu_Select()


atom response


response = message_box(
                          "Select a file to view from the 'Open' menu.\n"

                      & "The file is then displayed in the edit zone.", 

  "Help Menu",MB_OK)

end procedure

  onClick[AidMenu] = routine_id( "onAidMenu_Select" )

global procedure onAboutMenu_Select()


atom response


response = message_box(
                              "Illustration to finish Chapter 7 of my book",

      "This Program",MB_OK)

end procedure

  onClick[AboutMenu] = routine_id( "onAboutMenu_Select" )

WinMain(MainWindow,Normal)
The program could be improved in a number of respects, for example, it could monitor page length in the printer routine. I hope, however, it will give you a good guide as to how a working windows program is not too far away!
Postscript

I thought I would end with some guidance about what to do next. But first I would like to invite feedback from you, my audience. Please tell me if there are parts you find difficult or where more explanation would be helpful. By all means tell me if there are places where it is too easy or too detailed. I would also be interested to know, as your skills develop, what other features you would like in an introduction such as this.
Obviously there is much more that could be said. Some is about general programming techniques. If this is your first exposure to programming then I would certainly recommend that you look at the “Recent User Contributions” page on the Euphoria website, download any programs that sound interesting and study how the writer addressed his task. Not only will you learn about programming style (quite a personal thing) but also about the way in which programmers of Euphoria utilise (even exploit) the language.

The Euphoria Programming Language Reference Manual is not too daunting and is an essential read, albeit not page-by-page, in order to cover the areas I have not.
Unless you have a particular wish to write for the Command Prompt window, I would encourage you to work in a windowing environment. Try writing some more code using an editor until you feel you are getting the hang of it. Curiously, some of the things which are easiest in DOS are hardest in Windows, so don’t despair! By now there are lots of contributions using the win32lib library, so you can do some copying and editing of extant material.

At some stage I think you should try to use a different tool for writing windows program. Judith Evans, with lots of helpers, has constructed an IDE (Integrated Development Environment) which incorporates a Form Designer. The IDE opens with a blank window onto which you can paste as many widgets as you need. Each object can be given specific properties (Caption, position, dimensions, etc) and by double-clicking on an object a procedure editor is opened ready to accept your coding instructions for the particular event (click, right-click, scroll, close, etc) you are interested in. Programs written using the IDE are formed into Projects (a topic well beyond the scope of an introduction) but you can still look at an “exw” file and see how the overall program is built up from the elements in the IDE.
Details of how to download the IDE can be found on the “Recent User Contributions” page – try putting “Judith” in the search engine. It comes with some documentation which will help you on your way.

Good programming, and Good luck!

